#include <ESP8266WiFi.h>        // 引入ESP8266WiFi库，用于WiFi连接功能
#include <ESP8266WiFiMulti.h>   // 引入ESP8266WiFiMulti库，可以同时连接多个WiFi网络
#include <ESP8266WebServer.h>   // 引入ESP8266WebServer库，用于创建Web服务器

ESP8266WiFiMulti wifiMulti;     // 建立ESP8266WiFiMulti对象,对象名称是 'wifiMulti',用于管理WiFi连接

ESP8266WebServer esp8266_server(80);// 建立网络服务器对象，该对象用于响应HTTP请求。监听端口（80）

void add_user_AP(void) {
  //通过addAp函数存储  WiFi名称       WiFi密码
  // 这里可以添加多个WIFI wifiMulti会自动加载信号最强的，可以简化连接流程 提高连接成功率
  // 从而提高连接的可靠性和可用性
  wifiMulti.addAP("realme", "12345678");  // 使用addAP函数添加WiFi网络信息
  wifiMulti.addAP("real", "12345678");
  wifiMulti.addAP("me", "12345678");
}

// 等待连接成功
void wait_connect_ok(void) {
  Serial.println("正在连接Wifi,请稍等：");
  int i = 0;
  while (wifiMulti.run() != WL_CONNECTED) {  // 此处的wifiMulti.run()是重点。通过wifiMulti.run()，NodeMCU将会在当前
    delay(1000);                             // 环境中搜索addAP函数所存储的WiFi。如果搜到多个存储的WiFi那么NodeMCU
    Serial.print(i++); Serial.print(' ');    // 将会连接信号最强的那一个WiFi信号。
  }// 一旦连接WiFI成功，wifiMulti.run()将会返回“WL_CONNECTED”。这也是
  // 此处while循环判断是否跳出循环的条件。
}

// 打印连接信息
void print_connect_info(void) {
  // WiFi连接成功后将通过串口监视器输出连接成功信息
  Serial.println('\n');
  Serial.print("Connected to ");
  Serial.println(WiFi.SSID());              // 通过串口监视器输出连接的WiFi名称
  Serial.print("IP address:\t");
  Serial.println(WiFi.localIP());           // 通过串口监视器输出ESP8266-NodeMCU的IP
}

//启动网络服务功能
void start_web_server(void) {
  esp8266_server.begin();                           // 启动网站服务
  esp8266_server.on("/", HTTP_GET, handleRoot);     // 设置服务器根目录即'/'的函数'handleRoot'
  esp8266_server.on("/LED", HTTP_POST, handleLED);  // 设置处理LED控制请求的函数'handleLED'
  esp8266_server.onNotFound(handleNotFound);        // 设置处理404情况的函数'handleNotFound'

  Serial.println("HTTP esp8266_server started");//  告知用户ESP8266网络服务功能已经启动
}

// 初始化函数 用于初始化环境 只执行一次
void setup(void) {
  Serial.begin(115200);   // 启动串口通讯,115200波特率

  pinMode(5, OUTPUT); //设置内置LED引脚为输出模式以便控制LED
  //可使用这条语句初始化5号页脚为高电平(灯亮) 默认为低电平
  //digitalWrite(5, HIGH);

  add_user_AP(); // 添加几个可用的ap，WiFi重启后，自动连接信号最强的

  wait_connect_ok();// 等待连接成功

  print_connect_info();// 打印连接信息

  start_web_server();  // 启动网络服务功能
}

// 循环函数 初始化完成后重复执行 主要执行程序的主要逻辑代码
void loop(void) {
  // 监听、处理和响应客户端的HTTP请求
  esp8266_server.handleClient();  // 检查http服务器访问
}

// 拼接显示网页
String show_html(bool station)
{
    String html = "<!DOCTYPE html>";
    html += "<html lang =\"zh-cn\">";
    html += "<head>";
    html += "<meta charset=\"UTF-8\">";
    html += "<title>Wifi LED Light</title>";
    html += "<style>";
    html += "* {padding: 0;margin: 0;vertical-align: top;}";
    html += "div {margin: 10px auto;width: 100px;}";
    if (!station)
        html += "span {display: inline-block;margin: 2px;background-color:blue;width: 400px;height: 400px;top: 50%;left: 50%;position: absolute;transform: translate(-50%, -50%);border-radius: 50%;}";
    else
        html += "span {display: inline-block;margin: 2px;width: 400px;height: 400px;top: 50%;left: 50%;position: absolute;transform: translate(-50%, -50%);border-radius: 50%;}";
    html += "input {width: 300px;height: 300px;top: 50%;left: 50%;position: absolute;transform: translate(-50%, -50%);border-radius: 50%;}";
    html += "</style>";
    html += "</head>";
    html += "<body>";
    html += "<form action=\"/LED\" method=\"post\">";
    html += "<div><span></span><input type=\"submit\" value=\"LED\"></div>";
    html += "</form>";
    html += "</body>";
    html += "</html>";
    return html;
}


/*设置服务器根目录即'/'的函数'handleRoot'
  该函数的作用是每当有客户端访问NodeMCU服务器根目录时，
  NodeMCU都会向访问设备发送 HTTP 状态 200 (Ok) 这是send函数的第一个参数。
  同时NodeMCU还会向浏览器发送HTML代码，以下示例中send函数中第三个参数，
  也就是双引号中的内容就是NodeMCU发送的HTML代码。该代码可在网页中产生LED控制按钮。
  当用户按下按钮时，浏览器将会向NodeMCU的/LED页面发送HTTP请求，请求方式为POST。
  NodeMCU接收到此请求后将会执行handleLED函数内容*/
void handleRoot() {
  String html = show_html(digitalRead(5));//处理网站根目录“/”的访问请求
  //输出点平状态
  Serial.print("电平状态：");
  Serial.printlnm(digitalRead(5));
  esp8266_server.send(200, "text/html", html); // ESP8266开发板将调用此函数。
}

//处理LED控制请求的函数'handleLED' 
//改变页脚的返回值并重定向返回首页
void handleLED() {
  digitalWrite(5, !digitalRead(5)); // 改变LED的点亮或者熄灭状态 digitalRead将返回最后一次写入该引脚的值
  esp8266_server.sendHeader("Location", "/");         // 跳转回页面根目录
  esp8266_server.send(303);                           // 发送Http相应代码303 跳转
}

// 设置处理404情况的函数'handleNotFound'
void handleNotFound() {
  esp8266_server.send(404, "text/plain", "404: Not found"); // 发送 HTTP 状态 404 (未找到页面) 并向浏览器发送文字 "404: Not found"
}